Application Demonstration

Heating an Oil Stream Through Steam-Heated Tanks

© Maplesoft, a division of Waterloo Maple Inc., 2004

Introduction

Oil is to be heated by being pumped through a series of steam-heated tanks. This application shows how Maple can be used to design and analyze such a system.

In particular, we will derive closed-form formulas for the transient and steady-state temperature of the oil in each tank as a function of time, in terms of the flow rate of the oil (*W*), the specific heat capacity of the oil (*Cp*), the mass of the oil in each tank (*M*), the heat transfer coefficient (*U*), the heat transfer area (*A*), the temperature of the cold oil (*T0*) and the temperature of the steam (*Ts*). We will also verify the dimensions of these formulas to get a quick check on the accuracy of the solution.

The following Maple techniques are highlighted:

- Symbolic solving of systems of ordinary differential equations
- Management of units and dimensions
- Manipulation of symbolic expressions to yield further insights into a problem
- Deployment of the solution as an Excel application through automatic generation of Visual Basic code
- Deployment of the solution as a Maplet application

Initialization

```
> restart:
  interface( warnlevel = 0 ):
  with( CodeGeneration ):
```

Model

Below is a schematic of the heating system. Cold oil is pumped through three identical steam-heated tanks at a constant rate.



Standard texts on chemical engineering show that the temperature of the oil in each tank can be modeled by the following system of differential equations:

$$M\,C_p\,\frac{d}{dt}T_1(t) = \mathrm{W}\,C_p(T_0(t) - T_1(t)) + U\,\mathrm{A}(T_s - T_1(t))$$

$$M\,C_p\,\frac{d}{dt}T_2(t) = \mathrm{W}\,C_p(T_1(t) - T_2(t)) + U\,\mathrm{A}(T_s - T_2(t))$$

$$M\,C_p\,\frac{d}{dt}T_3(t) = \mathrm{W}\,C_p(T_2(t) - T_3(t)) + U\,\mathrm{A}(T_s - T_3(t))$$

where the parameters are defined as follows:

W	*Flow rate of the oil*	$\frac{kg}{s}$
M	*Mass of oil in each tank*	kg
C_p	*Specific heat capacity*	$\frac{J}{kg\,K}$
U	*Heat transfer coefficient*	$\frac{W}{m^2\,K}$
A	*Heat transfer area*	m^2
T_0	*Temperature of cold oil stream*	C
$T_1(t)$ $T_2(t)$ $T_3(t)$	*Transient temperatures*	C
T_s	*Temperature of steam*	C

We first enter this model into Maple

Temperature in Tank 1

```
> eq1 :=  M*Cp * diff(T1(t),t) = W*Cp*(T0-T1(t)) + U*A*(Ts-
T1(t));
```

$$eq1 := M\,Cp\left(\frac{d}{dt}\mathrm{T1}(t)\right) = \mathrm{W}\,Cp\,(T0 - \mathrm{T1}(t)) + U\,A\,(Ts - \mathrm{T1}(t))$$

Temperature in Tank 2

```
> eq2 := M*Cp * diff(T2(t),t) = W*Cp*(T1(t)-T2(t))+U*A*(Ts-
T2(t));
```

$$eq2 := M\,Cp\left(\frac{d}{dt}\mathrm{T2}(t)\right) = \mathrm{W}\,Cp\,(\mathrm{T1}(t) - \mathrm{T2}(t)) + U\,A\,(Ts - \mathrm{T2}(t))$$

Temperature in Tank 3

```
> eq3 := M*Cp * diff(T3(t),t) = W*Cp*(T2(t)-T3(t))+U*A*(Ts-
T3(t));
```

$$eq3 := M\,Cp\left(\frac{d}{dt}\mathrm{T3}(t)\right) = \mathrm{W}\,Cp\,(\mathrm{T2}(t) - \mathrm{T3}(t)) + U\,A\,(Ts - \mathrm{T3}(t))$$

Initial conditions

```
> Initial := T1(0) = T0, T2(0) = T0, T3(0) = T0;
```

$$Initial := \mathrm{T1}(0) = T0,\ \mathrm{T2}(0) = T0,\ \mathrm{T3}(0) = T0$$

Solve

We can now easily solve the model equations symbolically in one step.

```
> sols := dsolve( {eq1, eq2, eq3, Initial}, parametric );
```

$$
sols := \Bigg\{ \mathrm{T3}(t) = \Bigg(3\,U\,A\,Ts\,\mathrm{W}^2\,Cp^2\,\mathrm{e}^{\left(\frac{t\,\mathrm{W}}{M}\right)}\,\mathrm{e}^{\left(\frac{t\,U\,A}{M\,Cp}\right)} + 3\,U^2\,A^2\,Ts\,\mathrm{W}\,Cp\,\mathrm{e}^{\left(\frac{t\,\mathrm{W}}{M}\right)}\,\mathrm{e}^{\left(\frac{t\,U\,A}{M\,Cp}\right)}
$$

$$
+ \mathrm{W}^3\,Cp^3\,T0\,\mathrm{e}^{\left(\frac{t\,\mathrm{W}}{M}\right)}\,\mathrm{e}^{\left(\frac{t\,U\,A}{M\,Cp}\right)} + U^3\,A^3\,Ts\,\mathrm{e}^{\left(\frac{t\,\mathrm{W}}{M}\right)}\,\mathrm{e}^{\left(\frac{t\,U\,A}{M\,Cp}\right)} - 3\,\mathrm{W}\,Cp\,U^3\,A^3\,(3\,\mathrm{W}^2\,Ts\,Cp^2
$$

$$
+ 3\,\mathrm{W}\,U\,A\,Ts\,Cp + U^2\,A^2\,Ts - 3\,T0\,\mathrm{W}\,Cp\,U\,A - T0\,U^2\,A^2 - 3\,\mathrm{W}^2\,T0\,Cp^2) \big/ (
$$

$$
3\,\mathrm{W}\,Cp\,U^2\,A^2 + U^3\,A^3 + \mathrm{W}^3\,Cp^3 + 3\,\mathrm{W}^2\,Cp^2\,U\,A) - U^4\,A^4\,(3\,\mathrm{W}^2\,Ts\,Cp^2
$$

$$
+ 3\,\mathrm{W}\,U\,A\,Ts\,Cp + U^2\,A^2\,Ts - 3\,T0\,\mathrm{W}\,Cp\,U\,A - T0\,U^2\,A^2 - 3\,\mathrm{W}^2\,T0\,Cp^2) \big/ (
$$

$$
3\,\mathrm{W}\,Cp\,U^2\,A^2 + U^3\,A^3 + \mathrm{W}^3\,Cp^3 + 3\,\mathrm{W}^2\,Cp^2\,U\,A) - \mathrm{W}^3\,Cp^3\,U\,A\,(3\,\mathrm{W}^2\,Ts\,Cp^2
$$

$$
+ 3\,\mathrm{W}\,U\,A\,Ts\,Cp + U^2\,A^2\,Ts - 3\,T0\,\mathrm{W}\,Cp\,U\,A - T0\,U^2\,A^2 - 3\,\mathrm{W}^2\,T0\,Cp^2) \big/ (
$$

$$
3\,\mathrm{W}\,Cp\,U^2\,A^2 + U^3\,A^3 + \mathrm{W}^3\,Cp^3 + 3\,\mathrm{W}^2\,Cp^2\,U\,A) - 3\,\mathrm{W}^2\,Cp^2\,U^2\,A^2\,(3\,\mathrm{W}^2\,Ts\,Cp^2
$$

$$
+ 3\,\mathrm{W}\,U\,A\,Ts\,Cp + U^2\,A^2\,Ts - 3\,T0\,\mathrm{W}\,Cp\,U\,A - T0\,U^2\,A^2 - 3\,\mathrm{W}^2\,T0\,Cp^2) \big/ (
$$

$$
3\,\mathrm{W}\,Cp\,U^2\,A^2 + U^3\,A^3 + \mathrm{W}^3\,Cp^3 + 3\,\mathrm{W}^2\,Cp^2\,U\,A) - \frac{3\,\mathrm{W}^3\,Cp\,U^3\,A^3\,(Ts - T0)\,t^2}{2\,M^2\,(\mathrm{W}\,Cp + U\,A)}
$$

$$
- \frac{U^4\,A^4\,\mathrm{W}^2\,(Ts - T0)\,t^2}{2\,M^2\,(\mathrm{W}\,Cp + U\,A)} - \frac{\mathrm{W}^5\,Cp^3\,U\,A\,(Ts - T0)\,t^2}{2\,M^2\,(\mathrm{W}\,Cp + U\,A)} - \frac{3\,\mathrm{W}^4\,Cp^2\,U^2\,A^2\,(Ts - T0)\,t^2}{2\,M^2\,(\mathrm{W}\,Cp + U\,A)}
$$

$$
- \frac{3\,t\,\mathrm{W}^2\,U^3\,A^3\,(2\,\mathrm{W}\,Ts\,Cp + U\,A\,Ts - 2\,\mathrm{W}\,Cp\,T0 - U\,A\,T0)\,Cp}{M\,(\mathrm{W}^2\,Cp^2 + 2\,\mathrm{W}\,Cp\,U\,A + U^2\,A^2)}
$$

$$
- \frac{t\,\mathrm{W}\,U^4\,A^4\,(2\,\mathrm{W}\,Ts\,Cp + U\,A\,Ts - 2\,\mathrm{W}\,Cp\,T0 - U\,A\,T0)}{M\,(\mathrm{W}^2\,Cp^2 + 2\,\mathrm{W}\,Cp\,U\,A + U^2\,A^2)}
$$

$$
- \frac{t\,\mathrm{W}^4\,U\,A\,(2\,\mathrm{W}\,Ts\,Cp + U\,A\,Ts - 2\,\mathrm{W}\,Cp\,T0 - U\,A\,T0)\,Cp^3}{M\,(\mathrm{W}^2\,Cp^2 + 2\,\mathrm{W}\,Cp\,U\,A + U^2\,A^2)}
$$

$$
- \frac{3\,t\,\mathrm{W}^3\,U^2\,A^2\,(2\,\mathrm{W}\,Ts\,Cp + U\,A\,Ts - 2\,\mathrm{W}\,Cp\,T0 - U\,A\,T0)\,Cp^2}{M\,(\mathrm{W}^2\,Cp^2 + 2\,\mathrm{W}\,Cp\,U\,A + U^2\,A^2)} \Bigg) \Bigg/ \Bigg(\mathrm{e}^{\left(\frac{t\,\mathrm{W}}{M}\right)}
$$

$$
\mathrm{e}^{\left(\frac{t\,U\,A}{M\,Cp}\right)}\,(3\,\mathrm{W}\,Cp\,U^2\,A^2 + U^3\,A^3 + \mathrm{W}^3\,Cp^3 + 3\,\mathrm{W}^2\,Cp^2\,U\,A) \Bigg), \mathrm{T1}(t) = \Bigg(\mathrm{W}^3\,Cp\,T0
$$

$$
+ \mathrm{W}^2\,U\,A\,Ts - \frac{\mathrm{W}^3\,U\,A\,(Ts - T0)\,Cp\,\mathrm{e}^{\left(-\frac{(\mathrm{W}\,Cp + U\,A)\,t}{M\,Cp}\right)}}{\mathrm{W}\,Cp + U\,A}
$$

$$
- \frac{U^2\,A^2\,\mathrm{W}^2\,(Ts - T0)\,\mathrm{e}^{\left(-\frac{(\mathrm{W}\,Cp + U\,A)\,t}{M\,Cp}\right)}}{\mathrm{W}\,Cp + U\,A} \Bigg) \Big/ ((\mathrm{W}\,Cp + U\,A)\,\mathrm{W}^2), \mathrm{T2}(t) = \Bigg(\mathrm{W}^3\,T0\,Cp^2
$$

$$+2\,W^2\,U\,A\,Ts\,Cp + U^2\,A^2\,Ts\,W - \frac{2\,W^3\,U^2\,A^2\,(Ts - T0)\,t\,Cp\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{M\,(W\,Cp + U\,A)}$$

$$- \frac{W^2\,U^3\,A^3\,(Ts - T0)\,t\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{M\,(W\,Cp + U\,A)} - \frac{W^4\,U\,A\,(Ts - T0)\,t\,Cp^2\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{M\,(W\,Cp + U\,A)}$$

$$- \frac{2\,W^2\,U^2\,A^2\,(2\,W\,Ts\,Cp + U\,A\,Ts - 2\,W\,Cp\,T0 - U\,A\,T0)\,Cp\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{W^2\,Cp^2 + 2\,W\,Cp\,U\,A + U^2\,A^2}$$

$$- \frac{U^3\,A^3\,W\,(2\,W\,Ts\,Cp + U\,A\,Ts - 2\,W\,Cp\,T0 - U\,A\,T0)\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{W^2\,Cp^2 + 2\,W\,Cp\,U\,A + U^2\,A^2}$$

$$\left. - \frac{W^3\,U\,A\,(2\,W\,Ts\,Cp + U\,A\,Ts - 2\,W\,Cp\,T0 - U\,A\,T0)\,Cp^2\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{W^2\,Cp^2 + 2\,W\,Cp\,U\,A + U^2\,A^2}\right) \Big/ \Big($$

$$\left.(W\,Cp + U\,A)^2\,W)\right\}$$

Test

Maple has given us the explicit formulas describing the transient change in temperature in each tank. It's now easier to investigate the natural consequences of the model equations.

Before we proceed, we should check that the formulas make dimensional sense. Maple provides built-in support for units management to make this task fast and easy.

Verify the units and dimensions

```
> t1 := eval(T1(t), sols);
  t2 := eval(T2(t), sols);
  t3 := eval(T3(t), sols);
  with( Units[Standard] ):
```

$$t1 := \left(W^3\,Cp\,T0 + W^2\,U\,A\,Ts - \frac{W^3\,U\,A\,(Ts - T0)\,Cp\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{W\,Cp + U\,A}\right.$$

$$\left. - \frac{U^2\,A^2\,W^2\,(Ts - T0)\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{W\,Cp + U\,A}\right) \Big/ \left((W\,Cp + U\,A)\,W^2\right)$$

$$t2 := \Big(W^3\,T0\,Cp^2 + 2\,W^2\,U\,A\,Ts\,Cp + U^2\,A^2\,Ts\,W$$

$$- \frac{2\,W^3\,U^2\,A^2\,(Ts - T0)\,t\,Cp\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{M\,(W\,Cp + U\,A)}$$

$$- \frac{W^2\,U^3\,A^3\,(Ts - T0)\,t\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{M\,(W\,Cp + U\,A)} - \frac{W^4\,U\,A\,(Ts - T0)\,t\,Cp^2\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{M\,(W\,Cp + U\,A)}$$

$$- \frac{2\,W^2\,U^2\,A^2\,(2\,W\,Ts\,Cp + U\,A\,Ts - 2\,W\,Cp\,T0 - U\,A\,T0)\,Cp\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{W^2\,Cp^2 + 2\,W\,Cp\,U\,A + U^2\,A^2}$$

$$- \frac{U^3\,A^3\,W\,(2\,W\,Ts\,Cp + U\,A\,Ts - 2\,W\,Cp\,T0 - U\,A\,T0)\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{W^2\,Cp^2 + 2\,W\,Cp\,U\,A + U^2\,A^2}$$

$$- \left.\frac{W^3\,U\,A\,(2\,W\,Ts\,Cp + U\,A\,Ts - 2\,W\,Cp\,T0 - U\,A\,T0)\,Cp^2\,e^{\left(-\frac{(W\,Cp + U\,A)\,t}{M\,Cp}\right)}}{W^2\,Cp^2 + 2\,W\,Cp\,U\,A + U^2\,A^2}\right) \Big/ ($$

$$(W\,Cp + U\,A)^2\,W)$$

$$t3 := \left(3\,U\,A\,Ts\,W^2\,Cp^2\,e^{\left(\frac{t\,W}{M}\right)}\,e^{\left(\frac{t\,U\,A}{M\,Cp}\right)} + 3\,U^2\,A^2\,Ts\,W\,Cp\,e^{\left(\frac{t\,W}{M}\right)}\,e^{\left(\frac{t\,U\,A}{M\,Cp}\right)}\right.$$

$$+ W^3\,Cp^3\,T0\,e^{\left(\frac{t\,W}{M}\right)}\,e^{\left(\frac{t\,U\,A}{M\,Cp}\right)} + U^3\,A^3\,Ts\,e^{\left(\frac{t\,W}{M}\right)}\,e^{\left(\frac{t\,U\,A}{M\,Cp}\right)} - 3\,W\,Cp\,U^3\,A^3\,(3\,W^2\,Ts\,Cp^2$$

$$+ 3\,W\,U\,A\,Ts\,Cp + U^2\,A^2\,Ts - 3\,T0\,W\,Cp\,U\,A - T0\,U^2\,A^2 - 3\,W^2\,T0\,Cp^2) \Big/ ($$

$$3\,W\,Cp\,U^2\,A^2 + U^3\,A^3 + W^3\,Cp^3 + 3\,W^2\,Cp^2\,U\,A) - U^4\,A^4\,(3\,W^2\,Ts\,Cp^2$$

$$+ 3\,W\,U\,A\,Ts\,Cp + U^2\,A^2\,Ts - 3\,T0\,W\,Cp\,U\,A - T0\,U^2\,A^2 - 3\,W^2\,T0\,Cp^2) \Big/ ($$

$$3\,W\,Cp\,U^2\,A^2 + U^3\,A^3 + W^3\,Cp^3 + 3\,W^2\,Cp^2\,U\,A) - W^3\,Cp^3\,U\,A\,(3\,W^2\,Ts\,Cp^2$$

$$+ 3\,W\,U\,A\,Ts\,Cp + U^2\,A^2\,Ts - 3\,T0\,W\,Cp\,U\,A - T0\,U^2\,A^2 - 3\,W^2\,T0\,Cp^2) \Big/ ($$

$$3\,W\,Cp\,U^2\,A^2 + U^3\,A^3 + W^3\,Cp^3 + 3\,W^2\,Cp^2\,U\,A) - 3\,W^2\,Cp^2\,U^2\,A^2\,(3\,W^2\,Ts\,Cp^2$$

$$+ 3\,W\,U\,A\,Ts\,Cp + U^2\,A^2\,Ts - 3\,T0\,W\,Cp\,U\,A - T0\,U^2\,A^2 - 3\,W^2\,T0\,Cp^2) \Big/ ($$

$$3\,W\,Cp\,U^2\,A^2 + U^3\,A^3 + W^3\,Cp^3 + 3\,W^2\,Cp^2\,U\,A) - \frac{3\,W^3\,Cp\,U^3\,A^3\,(Ts - T0)\,t^2}{2\,M^2\,(W\,Cp + U\,A)}$$

$$- \frac{U^4\,A^4\,W^2\,(Ts - T0)\,t^2}{2\,M^2\,(W\,Cp + U\,A)} - \frac{W^5\,Cp^3\,U\,A\,(Ts - T0)\,t^2}{2\,M^2\,(W\,Cp + U\,A)} - \frac{3\,W^4\,Cp^2\,U^2\,A^2\,(Ts - T0)\,t^2}{2\,M^2\,(W\,Cp + U\,A)}$$

$$- \frac{3\,t\,W^2\,U^3\,A^3\,(2\,W\,Ts\,Cp + U\,A\,Ts - 2\,W\,Cp\,T0 - U\,A\,T0)\,Cp}{M\,(W^2\,Cp^2 + 2\,W\,Cp\,U\,A + U^2\,A^2)}$$

$$- \frac{t\,W\,U^4\,A^4\,(2\,W\,Ts\,Cp + U\,A\,Ts - 2\,W\,Cp\,T0 - U\,A\,T0)}{M\,(W^2\,Cp^2 + 2\,W\,Cp\,U\,A + U^2\,A^2)}$$

$$- \frac{t\, W^4\, U\, A\, (2\, W\, Ts\, Cp + U\, A\, Ts - 2\, W\, Cp\, T0 - U\, A\, T0)\, Cp^3}{M\, (W^2\, Cp^2 + 2\, W\, Cp\, U\, A + U^2\, A^2)}$$

$$\left. - \frac{3\, t\, W^3\, U^2\, A^2\, (2\, W\, Ts\, Cp + U\, A\, Ts - 2\, W\, Cp\, T0 - U\, A\, T0)\, Cp^2}{M\, (W^2\, Cp^2 + 2\, W\, Cp\, U\, A + U^2\, A^2)} \right) \Bigg/ \left(e^{\left(\frac{t\, W}{M}\right)} \right.$$

$$\left. e^{\left(\frac{t\, U\, A}{M\, Cp}\right)} (3\, W\, Cp\, U^2\, A^2 + U^3\, A^3 + W^3\, Cp^3 + 3\, W^2\, Cp^2\, U\, A) \right)$$

Here are some sample values for the parameters *W, M, Cp, U, A, T0* and *Ts* **with units attached**.

```
> sampleValues := [
  W  = 100 * Unit(kg/s),
  M  = 1000 * Unit(kg),
  Cp = 2 * Unit(J/(kg*K)),
  U  = 10 * Unit(W/(m^2*K)),
  A  = 1 * Unit(m^2),
  T0 = 10 * Unit(degC),
  Ts = 250 * Unit(degC),
  t  = t * Unit(s) ];
```

$$sampleValues := \left[W = 100 \left[\frac{kg}{s}\right], M = 1000\, [kg], Cp = 2 \left[\frac{m^2}{s^2\, K}\right], U = 10 \left[\frac{kg}{s^3\, K}\right], \right.$$

$$\left. A = [m^2], T0 = 10\, [degC], Ts = 250\, [degC], t = t\, [s] \right]$$

Here is the formula for the temperature in Tank 1 evaluated at these sample values. **Note that the dimensions reduce to temperature (Kelvin (K)), as expected**. This gives us an extra measure of confidence that Maple's solution to the model equations is correct.

```
> T1(t) := eval( t1, sampleValues );
```

$$T1(t) := \left(\frac{150}{7} - \frac{80}{7}\, e^{\left(-\frac{21\, t}{200}\right)} \right) [K]$$

And the temperature in Tank 2:

```
> T2(t) := eval( t2, sampleValues );
```

$$T2(t) := \left(\frac{4750}{147} - \frac{8}{7}\, t\, e^{\left(-\frac{21\, t}{200}\right)} - \frac{3280}{147}\, e^{\left(-\frac{21\, t}{200}\right)} \right) [K]$$

And the temperature in Tank 3:

```
> T3(t) := eval( t3, sampleValues );
```

$$T3(t) := \frac{1}{9261000}\left(\frac{395250000\,e^{\left(\frac{t}{10}\right)}\,e^{\left(\frac{t}{200}\right)} - 302640000 - 529200\,t^2 - 20664000\,t}{e^{\left(\frac{t}{10}\right)}\,e^{\left(\frac{t}{200}\right)}}\,[K]\right)$$

```
> plot([convert(T1(t), unit_free), convert(T2(t),
  unit_free),convert(T3(t), unit_free)], t=0..60,
  color=[red,blue,black], thickness=2, labels=["Time",
  "Temperature (K)"], labeldirections=[HORIZONTAL,
  VERTICAL], legend=["Tank 1", "Tank 2", "Tank 3"],
  title="Transient Temperatures vs. Time");
```



Next we wish to see what happens in the steady state. That is, what temperature does the oil stream converge to when it is fully heated? Maple will again help us derive symbolic formulas that will yield further insight into the problem. In particular, **we will be able to predict the steady-state temperature in a system with *any number of tanks*, not just three**.

Steady-state temperature in Tank 1

We can extract an expression describing the steady-state temperature in Tank 1 by finding the limit of T1(t) as t approaches infinity assuming all parameters are positive.

```
> limit(t1, t = infinity) assuming positive:
convert(%, parfrac, U);
```

$$Ts - \frac{\mathrm{W}\, Cp\, (Ts - T0)}{\mathrm{W}\, Cp + U\, A}$$

We can find the steady-state temperature in the other two tanks following a similar process.

Steady-state temperature in Tank 2

```
> limit(t2, t = infinity) assuming positive:
convert(%, parfrac, U);
```

$$Ts - \frac{\mathrm{W}^2\, Cp^2\, (Ts - T0)}{(\mathrm{W}\, Cp + U\, A)^2}$$

Steady-state temperature in Tank 3

```
> limit(t3, t = infinity) assuming positive:
convert(%, parfrac, U);
```

$$Ts - \frac{\mathrm{W}^3\, Cp^3\, (Ts - T0)}{(\mathrm{W}\, Cp + U\, A)^3}$$

Steady-state temperature in Tank N

From the formulas above, we can infer that the steady-state temperature in the *n*th tank is:

$$Ts + \frac{\mathrm{W}^n\, Cp^n\, (-Ts + T0)}{(\mathrm{W}\, Cp + U\, A)^n}$$

This formula gives us considerable predictive power. We now know how a system of any number of tanks will perform in the steady state, without having to build any prototypes or do any more modeling.

Deploy

We have used Maple to solve the problem as stated. The final challenge is to put that solution into the hands of the people who can benefit from it. Maple provides numerous options for such deployment. We show two of them below.

Deployment as a Maplet application

First, we deploy this solution as a Maplet application. It provides a point-and-click interface in which users can experiment with how the system behaves in response to different parameter values. Users don't need to know any Maple commands.or even the theory behind the physical process.



Deployment to Excel using Visual Basic code generation

Next, we will use **Maple's automatic code generation** to convert the formulas to Visual Basic code. We can then copy and paste this code into an Excel macro, so that **anyone with Excel on their desktop can make use of the solution we have derived**.

Visual Basic code for $T1(t)$

```
> VisualBasic(simplify(t1), resultname = temp_1,
coercetypes= false, optimize);
t4 = W * Cp
t6 = U * A
t10 = t4 + t6
t13 = Exp(-t * t10 / M / Cp)
temp_1 = 0.1E1 / t10 * (T0 * t4 - Ts * t13 * t6 + T0 * t13
* t6 + Ts * t6)
```

*Visual Basic code for T*2(t)

```
> VisualBasic(simplify(t2), resultname = temp_2,
coercetypes= false, optimize);
t4 = W * W
t6 = Cp * Cp
t10 = U * t4 * Cp
t11 = A * t
t12 = 0.1E1 / M
t15 = W * Cp
t16 = U * A
t20 = Exp(-t * (t15 + t16) / Cp * t12)
t33 = U * t15
t34 = t20 * A
t35 = Ts * M
t43 = U * U
t45 = A * A
t46 = t45 * t43 * W
t47 = t20 * t
t52 = t45 * t43
t54 = M * t20
t59 = -M * t6 * T0 * t4 + Ts * t20 * t11 * t10 - T0 * t20 *
t11 * t10 - 2 * W * U * A * Ts * Cp * M + 2 * t35 * t34 *
t33 - 2 * M * T0 * t34 * t33 - T0 * t47 * t46 + Ts * t47 *
t46 - t35 * t52 + Ts * t54 * t52 - T0 * t54 * t52
temp_2 = -t12 / (t6 * t4 + 2 * t16 * t15 + t52) * t59
```

*Visual Basic code for T*3(t)

```
> VisualBasic(simplify(t3), resultname = temp_3,
coercetypes= false, optimize);
t7 = W * Cp
t8 = U * A
t9 = t7 + t8
t11 = t * t9 / M / Cp
t12 = Exp(-t11)
t13 = W * W
t14 = t13 * W
t15 = Cp * Cp
t18 = Exp(t11)
t20 = M * M
t24 = t13 * t13
t26 = U * t24 * t15
t27 = t * t
t28 = t27 * A
t34 = t * t14 * t15
t35 = M * T0
t39 = M * Ts
t43 = t13 * t15
```

```
t44 = U * t43
t45 = t20 * A
t50 = t20 * t18
t58 = U * U
t59 = t58 * t14 * Cp
t60 = A * A
t61 = t27 * t60
t69 = t * t13 * Cp
t70 = t60 * t58
t74 = -2 * t20 * t18 * T0 * t15 * Cp * t14 - T0 * t28 * t26
+ Ts * t28 * t26 - 4 * t35 * t8 * t34 + 4 * t39 * t8 * t34
+ 6 * Ts * t45 * t44 - 6 * t50 * t43 * Ts * t8 - 6 * T0 *
t45 * t44 - 2 * T0 * t61 * t59 + 2 * Ts * t61 * t59 - 6 *
t35 * t70 * t69
t82 = t58 * t7
t83 = t20 * t60
t90 = t58 * U
t91 = t90 * t13
t92 = t60 * A
t93 = t27 * t92
t99 = t90 * t * W
t100 = M * t92
t107 = t92 * t90
t118 = 6 * t39 * t70 * t69 - 6 * t50 * t7 * Ts * t70 + 6 *
Ts * t83 * t82 - 6 * T0 * t83 * t82 - T0 * t93 * t91 + Ts *
t93 * t91 - 2 * T0 * t100 * t99 + 2 * Ts * t100 * t99 - 2 *
t20 * t18 * Ts * t107 - 2 * T0 * t20 * t107 + 2 * Ts * t20
* t107
temp_3 = -0.1E1 / t9 / t20 / (t43 + 2 * t8 * t7 + t70) *
(t74 + t118) * t12 / 2
```

Legal Notice: The copyright for this application is owned by Maplesoft. The application is intended to demonstrate the use of Maplesoft products to solve a particular problem. It has been made available for product evaluation purposes only and may not be used in any other context without the express permission of Maplesoft.

Thank you for evaluating this Maple application sample www.maplesoft.com